

```

%macro md_clean_and_reset(
    debug          =N
    ,_workdata     =
    ,_workview     =
    ,_workcat      =
    ,resetmprint   =
    ,libname       =
    ) / store source des='V1.0.0.9' ;
%*****
*****
* FILENAME:      MD_CLEAN_AND_RESET.SAS
* DEVELOPER:     (b) (6)
* PLATFORM:      SAS 9.1.3, 9.2 on PC
*
* MACROS USED:   None
* ASSUMPTIONS:   if the DEBUG parm is not Y and values have been entered in _WORKDATA
and/or _WORKVIEW
*               those values are existing work data set names.
*
*               Please note that if MD_WORKINFO has been invoked prior to this macro
a list of existing work data sets
*               and/or work views have been created in global macro variables and
these can be used as values
*               to the respective macro paramters, along with any data sets created
after call to MD_WORKINFO,
*               in this macro.. see Usage notes for more info.
*
* DESCRIPTION:
* 1. This macro will set the MPRINT options value to whatever is passed
* 2. If DEBUG parameter is NOT set to Y, proc dataset processing is performed to
SAVE a selected set of
*       existing WORK data sets as entered in macro paramter _WORDDATA
*       and/or existing WORK data views as entered in macro paramter _WORKVIEW
*       Meaning all others will be deleted.
*
*
* USAGE NOTES:
* -- when
* %md_clean_and_reset(
*     debug          = &debug
*     ,_workdata     = %str(&WORK_DATASETS_DATA &out_data)
*     ,_workview     = %str(&WORK_DATASETS_VIEW)
*     ,_workcat      = %str(GSEG SASGOPT)
*     ,resetmprint   = &mprint_setting
*     )
*
* -- To remove all work data sets and views, and reset mprint option setting to
provided value
* %md_clean_and_reset
*

```

```

* PARAMETERS:
* DEBUG          = N (Default), Deletes all work data sets/views except those
passed into macro paramters:
*                _WORKDATA and _WORKVIEW
* _WORKDATA      = list of WORK data sets to be saved (not deleted) null value is
DEFAULT (Delete all work data sets)
* _WORKVIEW      = list of WORK data views to be saved (not deleted) null value is
DEFAULT (Delete all work data views)
* _WORKCAT       = list of WORK catalogs to be saved (not deleted) null value is
DEFAULT (Delete all work catalogs except FORMATS and SASMACR)
* RESETMPRINT    = MPRINT Option Setting to set in this macro , null value is
DEFAULT (Set to NOMPRINT)
* LIBNAME        = The library for which data sets and views will be saved.
*                The parameter value should be a a single unquoted token.
*                If the parameter value is null then reset it to WORK.
*****
* DATE          INITIALS   MODIFICATION DESCRIPTION
*****
*
*****
* © 2011 Pharmaceutical Product Development, Inc.
* All Rights Reserved.
*****
*****;

```

```

%if %upcase(%bquote(&debug)) ne Y %then %do;

    %local libname items;

    %if %bquote(&libname) = %str( ) %then %let libname = WORK;
    %else                                     %let libname = %upcase(&libname);

    proc sql noprint;
        select memname into :items
            separated by ' '
            from dictionary.members
            where libname = "&libname"
            and memtype = 'ITEMSTOR'
        ;
    quit;

    %if %md_notnull(_workcat) %then %do;
        %mu_wordscan(string= %str(&_workcat ), root=cat,    numw=ncats,
delim=%str( ) )
        %end;
    %else %let ncats = 0;

    %if %md_notnull(_workdata) %then %do;
        %mu_wordscan(string= %str(&_workdata ), root=data,    numw=ndata,

```

```

delim=%str( ) )
    %end;
    %else %let ndata = 0;

        %if %md_notnull(_workview) %then %do;
            %mu_wordscan(string= %str(&_workview ), root=view,    numw=nview,
delim=%str( ) )
        %end;
    %else %let nview = 0;

proc datasets lib = &libname nolist;

    %do i = 1 %to &ndata;
    %if %sysfunc(exist(&libname..&&data&i)) %then
        %str(save &&data&i / mt=data; ) ;
    %end;

    %if %sysfunc(cexist(&libname..formats)) %then
        %str(save formats / mt=catalog; ) ;

    %if %sysfunc(cexist(&libname..sasmacr)) %then
        %str(save sasmacr / mt=catalog; ) ;

    %if %bquote(&items) ne %str( ) %then
        %str(save &items / mt=itemstor; ) ;

    %do v = 1 %to &nview;
    %if %sysfunc(exist(&libname..&&view&v, view)) %then
        %str(save &&view&v / mt=view; ) ;
    %end;

    %do c = 1 %to &ncats;
        %if %sysfunc(cexist(&libname..&&cat&c)) %then
            %str(save &&cat&c / mt=catalog; ) ;
    %end;
quit;
%end;

%if %bquote(&resetmprint) = %str( ) %then %let resetmprint = NOMPRINT;
options &resetmprint ;
%mend md_clean_and_reset;

```